

## SYMBOL PLOT

### PURPOSE

Generates a symbol plot.

### DESCRIPTION

A symbol plot is a scatter plot for which the attributes of the plot symbols are controlled by the values of other variables. This plot allows the size, type, color, and fill attribute to be controlled.

### SYNTAX 1

SYMBOL PLOT <y> <x> <size> <SUBSET/EXCEPT/FOR qualification>

where <y> is the vertical axis variable;

<x> is the horizontal axis variable;

<size> is a variable that controls the size of the plot symbol;

and where the <SUBSET/EXCEPT/FOR qualification> is optional.

### SYNTAX 2

SYMBOL PLOT <y> <x> <size> <symbol> <SUBSET/EXCEPT/FOR qualification>

where <y> is the vertical axis variable;

<x> is the horizontal axis variable;

<size> is a variable that controls the size of the plot symbol;

<symbol> is a variable that controls the type of symbol plotted;

and where the <SUBSET/EXCEPT/FOR qualification> is optional.

### SYNTAX 3

SYMBOL PLOT <y> <x> <size> <symbol> <color> <SUBSET/EXCEPT/FOR qualification>

where <y> is the vertical axis variable;

<x> is the horizontal axis variable;

<size> is a variable that controls the size of the plot symbol;

<symbol> is a variable that controls the type of symbol plotted;

<color> is a variable that controls the color of the symbol;

and where the <SUBSET/EXCEPT/FOR qualification> is optional.

### SYNTAX 4

SYMBOL PLOT <y> <x> <size> <symbol> <color> <fill> <SUBSET/EXCEPT/FOR qualification>

where <y> is the vertical axis variable;

<x> is the horizontal axis variable;

<size> is a variable that controls the size of the plot symbol;

<symbol> is a variable that controls the type of symbol plotted;

<color> is a variable that controls the color of the symbol;

<fill> is an optional variable that controls whether the symbol is filled or not;

and where the <SUBSET/EXCEPT/FOR qualification> is optional.

### EXAMPLES

SYMBOL PLOT Y X PRESSURE

SYMBOL PLOT Y X PRESSURE TYPE

SYMBOL PLOT Y X PRESSURE TYPE ICOL

SYMBOL PLOT Y X PRESSURE TYPE ICOL IFILL

### NOTE 1

SYMBOL PLOT Y and SYMBOL PLOT Y X are equivalent to PLOT Y and PLOT Y X respectively. The only difference is that the line pattern is automatically set to blank.

### NOTE 2

The size for Y(i) is scaled to: Y(i)/MAXIMUM(ABS(Y)). The maximum value of Y is set to the size given in the CHARACTER SIZE (or CHARACTER HW) command while the others multiply this size by the scale factor. The minimum scale factor is truncated at 5%.

## NOTE 3

The <size>, <symbol>, <color>, and <fill> variables are optional, but must be entered in that order. To skip one, set all the elements of the skipped variable to 1. For example,

```
LET N = SIZE Y
LET SYMBOL = 1 FOR I = 1 1 N
SYMBOL PLOT Y X SIZE SYMBOL COLOR
```

## NOTE 4

The <symbol> and <color> variables define indices to the CHARACTER and CHARACTER COLOR commands respectively. That is, they contain values between 1 and 100 (e.g., a value of 3 would go to the third setting of these commands).

## NOTE 5

Zero values for the <fill> variable indicate non-filled characters while non-zero values indicate filled characters. In addition, to generate filled characters a fillable symbol must be used (e.g., CIRCLE or SQUARE).

## DEFAULT

None

## SYNONYMS

None

## RELATED COMMANDS

CHARACTER	=	Sets the type for plot characters.
CHARACTER COLOR	=	Sets the color for plot characters.
CHARACTER FILL	=	Sets the fill for plot characters.
ANDREWS PLOT	=	Generates an Andrews plot.
PROFILE PLOT	=	Generates a profile plot.
STAR PLOT	=	Generates a star plot.
PLOT	=	Generates a data or function plot.

## REFERENCE

“Graphical Methods for Data Analysis,” Chambers, Cleveland, Kleiner, and Tukey, Wadsworth, 1983.

## APPLICATIONS

Multivariate Analysis

## IMPLEMENTATION DATE

92/12

## PROGRAM

```

DIMENSION 20 COLUMNS
MULTIPLY 2 2; MULTIPLY CORNER COORDINATES 0 0 100 100
SKIP 25
READ CURRIE.DAT ID1 ID2 FRWC CTOTAL POT LEAD IRON
LET TYPE = ID1; LET TYPE = 1 SUBSET ID1 < 10
LET TYPE = 2 SUBSET ID1 = 11 TO 19; LET TYPE = 3 SUBSET ID1 > 19
LET N = SIZE ID1; LET DUMMY = 1 FOR I = 1 1 N
TITLE AUTOMATIC; Y1LABEL POTASSIUM; X1LABEL LEAD
CHARACTER HW 5.0 3.75 ALL

.
CHARACTER CIRCLE SQUARE DIAMOND TRIANGLE
LEGEND 1 IRON DETERMINES SYMBOL SIZE
SYMBOL PLOT POT LEAD IRON

.
CHARACTER HW 2.0 1.5 ALL
LEGEND 1 ID1 DETERMINES SYMBOL TYPE
SYMBOL PLOT POT LEAD DUMMY TYPE

.
CHARACTER HW 5.0 3.75 ALL
LEGEND 1 IRON DETERMINES SYMBOL SIZE
LEGEND 2 ID1 DETERMINES SYMBOL TYPE
SYMBOL PLOT POT LEAD IRON TYPE

.
LET AFILL = 0 FOR I = 1 1 N
LET AFILL = 1 SUBSET ID2 = 1
LEGEND 3 ID2 DETERMINES FILL
SYMBOL PLOT POT LEAD IRON TYPE DUMMY AFILL
END OF MULTIPLY

```

